1 Introduction

Guillermo Ortiz called our attention recently to Tamm modes, surface modes that exist at the interface between an homogeneous medium or vacuum and a crystal, even a 1D crystal with periodicity along the direction normal to the interface. We do a first exercise obtaining the dispersion relation of these modes.

2 Theory

2.1 System

We consider a system such as that in Fig. 1 made of a semiinfinite superlattice consisting of alternating layers $a$, $b$ of dielectrics with corresponding dielectric response $\epsilon_a$, $\epsilon_b$ and width $d_a$, $d_b$. The system is illuminated from below by a wave with wavevector $k_0 = (Q, k_0)$, where $Q$ lies on the interface and

$$k_0^2 = \epsilon_0 \frac{\omega^2}{c^2} - Q^2,$$

with $\epsilon_0$ the response of the ambient medium. There is also a reflected wave with wavevector $k_r = (Q, -k_0)$. Within each medium there are upgoing and downgoing waves with wavevectors $k_{\alpha \pm} = (Q, \pm k_\alpha)$ with

$$k_{\alpha}^2 = \epsilon_\alpha \frac{\omega^2}{c^2} - Q^2, \quad (\alpha = a, b)$$

2.2 Reflection amplitude

The reflection amplitude of the system may be obtained from a surface impedance formalism. Without loss of generality, we assume the interfaces
Figure 1: System consisting of a periodic superlattice made of two alternating dielectrics with response $\epsilon_a$ and $\epsilon_b$ with corresponding widths $d_a$ and $d_b$. Within each media there are upwards and downwards propagating waves with wavevectors $k_{\pm} = (Q, k_{\pm})$. 
are parallel to the $xy$ plane and that $Q = (Q, 0, 0)$ points along the $y$ direction. We separate the cases of TE ($s$) and TM ($p$) polarization. For $s$ polarization, the electric field would point along $x$ for all the waves. Thus, we define the reflection amplitude $r_s$ through

$$r_s = \frac{E_x^r}{E_x^i},$$

(3)

where $E^i$ and $E^r$ denote the amplitudes of the incident and reflected fields. Similarly, for $p$-polarization the magnetic field $H$ would point along $x$, suggesting we define the reflection amplitude $r_p$ through

$$r_p = \frac{H_x^r}{H_x^i},$$

(4)

We define the surface impedance of a medium as the quotient

$$Z = \frac{E\parallel}{H\parallel},$$

(5)

where $E\parallel$ and $H\parallel$ are the projections of the amplitude of the wave onto the surface. Thus, for $s$ polarization we chose $E\parallel = E_x$ and $H\parallel = H_y$, while for $p$ polarization we chose $E\parallel = -E_y$ and $H\parallel = H_y$. Then, using Faraday’s law and Ampere-Maxwell’s law we obtain the impedance for each medium

$$Z^{s}_\alpha = \frac{q}{k^{s}_\alpha}, \quad Z^{p}_\alpha = \frac{k^{p}_\alpha}{q\epsilon^{s}_\alpha}, \quad (\alpha = 0, a, b)$$

(6)

Notice that for downwards moving waves these impedances would change sign. Thus, for fields at the surface are given by total fields at the surface are related through

$$E_x^s = (1 + r^s)E_x^i, \quad H_y^s = (1 - r^s)E_x^i/Z_0^s,$$

$$E_y^p = -(1 - r^p)Z_0^pH_x^i, \quad H_x^p = (1 + r^p)H_x^i,$$

(7)

from which we obtain a relation between the surface impedances of the system $(Z^s = E_x^s/H_y^s, Z_p = -E_y^p/H_x^p)$ and the reflection amplitude

$$Z^s = \frac{1 + r^s}{1 - r^s}Z_0^s,$$

$$Z^p = \frac{1 - r^p}{1 + r^p}Z_0^p,$$

(8)
from which we obtain the reflection amplitudes

\[ r^s = \frac{Z_0^s - Z^s}{Z_0^s + Z^s}, \]
\[ r^p = \frac{Z^p - Z_0^p}{Z^p + Z_0^p}. \] (9)

The sign difference between the last two expressions is due to our definition of \( r^p \) in terms of the magnetic instead of the electric field amplitudes.

### 2.3 Normal modes

The normal modes of a system exist whenever the system has a response without being excited by an incoming external field. Correspondingly, the surface modes are given by the poles of the reflection amplitude. This means that the dispersion relation of the surface modes is given implicitly through the relations

\[ Z_0^s + Z^s = 0, \]
\[ Z_0^p + Z^p = 0. \] (10)

### 2.4 Transfer matrix

In order to apply Eqs. (10), we have to calculate the surface impedances \( Z^s \) and \( Z^p \) of the semiinfinite superlattice. To that end we first obtain the transfer matrix of a single layer. As the fields within one homogeneous layer is given by the sum of an upwards and a downwards moving wave, each characterized by a perpendicular wavenumber \( (\pm k_\alpha) \), field \( (E_{\alpha \pm}^s(z) \) for \( s \) and \( H_{\alpha \pm}^p(z) \) for \( p \) polarization) and surface impedance \( \pm Z_\alpha \). The parallel components of the electric and magnetic fields \( E_{\alpha \parallel}(z) \) and \( H_{\alpha \parallel}(z) \) are related to the upgoing and downgoing fields through

\[ E_{\alpha \parallel}^s(z) = E_{\alpha +}^s(z) + E_{\alpha -}^s(z), \]
\[ H_{\alpha \parallel}^s(z) = (E_{\alpha +}^s(z) - E_{\alpha -}^s(z)) Y_{\alpha}^s, \]
\[ E_{\alpha \parallel}^p(z) = (H_{\alpha +}^p(z) - H_{\alpha -}^p(z)) Z_{\alpha}^p, \]
\[ H_{\alpha \parallel}^p(z) = (H_{\alpha +}^p(z) + H_{\alpha -}^p(z)), \] (11)
which we write as the matrix equations

\[
\begin{pmatrix}
E^s_{\alpha} \\
H^s_{\alpha}
\end{pmatrix}_z = \begin{pmatrix}
1 & 1 \\
Y^s_{\alpha} & -Y^s_{\alpha}
\end{pmatrix}
\begin{pmatrix}
E^s_{\alpha+} \\
E^s_{\alpha-}
\end{pmatrix}_z,
\]

\[
\begin{pmatrix}
E^p_{\alpha} \\
H^p_{\alpha}
\end{pmatrix}_z = \begin{pmatrix}
Z^p_{\alpha} & -Z^p_{\alpha} \\
1 & 1
\end{pmatrix}
\begin{pmatrix}
E^p_{\alpha+} \\
E^p_{\alpha-}
\end{pmatrix}_z,
\]

(12)

where we introduced the surface admittance

\[Y^\zeta_{\alpha} \equiv 1/Z^\zeta_{\alpha}\]

for each polarization \(\zeta = s, p\). Together with

\[E^\zeta_{\alpha}(z) = E^\zeta_{\alpha}(z')e^{\pm ik_{\alpha}(z-z')}\]

for any two positions within the layer, allows us to write the fields at the upper interface of a layer \((u)\) in terms of the fields at its lower interface \((l)\),

\[
\begin{pmatrix}
E^\zeta_{\alpha} \\
H^\zeta_{\alpha}
\end{pmatrix}_u = M^\zeta_{\alpha}
\begin{pmatrix}
E^\zeta_{\alpha} \\
H^\zeta_{\alpha}
\end{pmatrix}_l
\]

(13)

where

\[
M^s_{\alpha} = \begin{pmatrix}
1 & 1 \\
Y^s_{\alpha} & -Y^s_{\alpha}
\end{pmatrix}
\begin{pmatrix}
1 & 0 \\
e^{ik_{alpha}d_{alpha}} & e^{-ik_{alpha}d_{alpha}}
\end{pmatrix}
\begin{pmatrix}
1 & 1 \\
Y^s_{\alpha} & -Y^s_{\alpha}
\end{pmatrix}^{-1},
\]

\[
M^p_{\alpha} = \begin{pmatrix}
Z^p_{\alpha} & -Z^p_{\alpha} \\
1 & 1
\end{pmatrix}
\begin{pmatrix}
e^{ik_{alpha}d_{alpha}} & 0 \\
e^{-ik_{alpha}d_{alpha}} & 0
\end{pmatrix}
\begin{pmatrix}
Z^p_{\alpha} & -Z^p_{\alpha} \\
1 & 1
\end{pmatrix}^{-1},
\]

(14)

yielding

\[
M^\zeta_{\alpha} = \begin{pmatrix}
\cos k_{\alpha}d_{\alpha} & iZ^\zeta_{\alpha} \sin k_{\alpha}d_{\alpha} \\
iY^\zeta_{\alpha} \sin k_{\alpha}d_{\alpha} & \cos k_{\alpha}d_{\alpha}
\end{pmatrix},
\]

(15)

which has the same form for both polarizations. From the transfer matrix of each layer, we may obtain the transfer matrix of a period as a simple matrix multiplication,

\[
M^\zeta = M^\zeta_{b} M^\zeta_{a},
\]

(16)

as the parallel fields \(E_\parallel\) and \(H_\parallel\) are continuous across each interface. Notice that by construction, the determinant of the transfer matrix is 1.

### 2.5 Bulk modes

Within a periodic system, the normal modes may be written as a superposition of Bloch modes, which acquire a phase \(e^{ipd}\) when translated across a period of width \(d = d_a + d_b\), where \(p\) is Bloch’s vector. Thus

\[
\begin{pmatrix}
E^\zeta_{\parallel} \\
H^\zeta_{\parallel}
\end{pmatrix}_d = M^\zeta
\begin{pmatrix}
E^\zeta_{\parallel} \\
H^\zeta_{\parallel}
\end{pmatrix}_0 = e^{ipd} \begin{pmatrix}
E^\zeta_{\parallel} \\
H^\zeta_{\parallel}
\end{pmatrix}_0,
\]

(17)
i.e., $e^{ipd}$ is an eigenvalue of the transfer matrix. Notice that for an infinite system, $p^c$ ought to be real in order to avoid divergences at $z = \pm\infty$. However, for semiinfinite systems we may also allow for complex Bloch’s vector as long as its imaginary part is positive, so that the mode decays as $z \to \infty$. The characteristic equation

$$\det(M^\zeta - \Lambda^\zeta) = (\Lambda^\zeta)^2 - \Lambda^\zeta \text{tr} M + \det M,$$

(18)

for the eigenvalue $\Lambda^\zeta = e^{ip^c d}$ may be simplified to

$$\cos p^c d = \frac{1}{2} \text{tr} M^\zeta,$$

(19)

as $\det M^\zeta = 1$.

As $\cos$ is an even function, for any solution $p$ there is also a solution $-p$. To identify the correct choice of a solution we may add a small imaginary part to the dielectric functions, to ensure that $p$ is complex, and choose that value that has a positive imaginary part.

### 2.6 Surface impedance

The surface impedance of the system may now be obtained from the eigenvector of $M^\zeta$,

$$(M^\zeta - \Lambda^\zeta) \begin{pmatrix} E^\zeta_\parallel \\ H^\zeta_\parallel \end{pmatrix},$$

(20)

from which we can read the surface impedance

$$Z^\zeta = \frac{M^\zeta_{12}}{\Lambda^\zeta - M^\zeta_{11}} = \frac{\Lambda^\zeta - M^\zeta_{22}}{M^\zeta_{21}}$$

$$= \pm \frac{M^\zeta_{12}}{M^\zeta_{22} - M^\zeta_{11} \pm \sqrt{4 - (M^\zeta_{11} + M^\zeta_{22})^2}}$$

$$= \pm \frac{1}{2} \frac{M^\zeta_{11} - M^\zeta_{22} \pm \sqrt{4 - (M^\zeta_{11} + M^\zeta_{22})^2}}{M^\zeta_{21}}.$$

(21)

The signs of the square roots should be chosen in such a way that the surface impedance has a positive real part, so that energy is dissipated (even if only infinitesimally for transparent systems) within the superlattice.
2.7 Summary

We have now all the ingredients for the calculation of the reflection amplitude and its poles, which yield the normal modes. For example, we could give $Q$ and $\omega$ and the polarization $\zeta$, calculate all wavevectors from Eqs. (1) and (2), the surface impedance of each medium through Eq. (6), the transfer matrix of each layer and the complete transfer matrix through Eqs. (15) and (16), Bloch’s vector through Eq. (19), the surface impedance of the superlattice from Eq. (21) and finally the reflection amplitude and/or dispersion relations from Eqs. (9) and (10).

3 Implementation

In this section we build a computational code to perform the calculations summarized above for a dispersionless dielectric system. We use the Perl Data Language PDL.

First we include the required pragmas and packages.

```perl
# Calculate and plot the reflection amplitude
# of a dielectric superlattice
use warnings;
use strict;
use v5.12;
use Getopt::Long;
use List::Util;
use PDL;
use PDL::NiceSlice;
use PDL::Graphics::Gnuplot;
```

Now we initialize the required parameters. I will use units such that the period is $d = 1$ and the speed of light is $c = 1$.

```perl
my $eps0=1;   # dielectric function of the ambient. Vacuum by default
my ($epsa, $epsb); # dielectric function of the layers
my ($da, $db);  # widths of layers a and b
my $im=1e-7;  # small imaginary part added to disambiguate
my $Qmin=0;   # minimum wavevector
my $Qmax;     # maximum wavevector
my $NQ;       # number of different wavevectors
my $wmin=0;   # minimum frequency
my $wmax;     # maximum frequency
```
my $Nw;  # number of frequencies
my $pol;  # polarization
my $filename;  # filename for plot
my $title='';  # title for plot
my $cbrange;
my $options=q(
  'eps0=f'=>\$eps0,  # dielectric function of ambient
  'epsa=f'=>\$epsa,  # dielectric function of layers a
  'epsb=f'=>\$epsb,  # dielectric function of layers b
  'im=f'=>\$im,  # imaginary part to add to epsilons
  'da=f'=>\$da,  # width of layers a
  'db=f'=>\$db,  # width of layers b, d-da
  'Qmin=f'=>\$Qmin,  # minimum wavevector
  'Qmax=f'=>\$Qmax,  # maximum wavevector
  'NQ=i'=>\$NQ,  # number of wavevectors
  'wmin=f'=>\$wmin,  # minimum frequency
  'wmax=f'=>\$wmax,  # maximum frequency
  'Nw=i'=>\$Nw,  # number of frequencies
  'pol=s'=>\$pol,  # polarization
  'filename=s'=>\$filename,  # filename for plot
  'title=s'=>\$title,  # title for plot
  'cbrange=s'=>\$cbrange,  # optional cbrange "[min,max]"
);
my %options=(eval $options);
die "Bad option definition; $@
if $@;
GetOptions(%options) or usage($options, "Bad options");
usage($options, "Undefined parameters")
unless List::Util::all
  {defined $_}
  ($eps0, $epsa, $epsb, $im, $da, $Qmin, $Qmax, $NQ,
   $wmin, $wmax, $Nw, $pol, $filename, $title);

We need a routine to print a usage message and quit when a mistake is
found.

sub usage {
  my ($options, $message)=@_;  
  say $message if defined $message;
  say $options;
  exit 1;
}
We check the input parameters and set the remaining parameter(s).

```perl
usage($options, "Parameter da should obey 0<=da<=1: $da")
unless 0<=$da<=1;
$db=1-$da; # enforce units so that d=d_a+d_b=1
my %pols=(te=>'s', s=>'s', p=>'p', tm=>'p');
my $polarization=$pols{lc $pol}; # normalize polarization
usage($options, "Polarization should be s, p, TE or TM: $pol")
unless defined $polarization;
my ($ceps0, $cepsa, $cepsb)
=map {$_+$im*i()} ($eps0, $epsa, $epsb); # add imaginary parts
```

Now we define the wavevector and frequency (Eqs. [1] and [2]) ndarrays and apply our formulas above. I use dummy arrays so I can thread over all wavevectors and frequencies.

```perl
my $Q=zeroes($NQ)->xlinvals($Qmin,$Qmax)->dummy(1); # Q,w
my $w=zeroes($Nw)->xlinvals($wmin,$wmax)->dummy(0); # Q,w
my $q=$w; # assume c=1
my ($k0, $ka, $kb)=map {mysqrt($_*$q**2-$Q**2)} ($ceps0, $cepsa, $cepsb);
```

I introduce a square root with branch cut in the negative real axis, so it yields a non-negative imaginary part. I use `thread_define` so that it can be threaded over arrays.

```perl
sub mysqrt {
    my $a=shift;
    my $b=null;
    _mysqrt($a, $b);
    return $b;
}
```

```perl
BEGIN{
    thread_define ',_mysqrt(a());[0]b())' => over {
        my ($a, $b)=@_;
        my $r=sqrt($a);
        $r= -$r if $r->im<0;
        $b.=$r;
    };
}
```

I calculate the surface impedance (Eqs. [6]),
my ($Z0, $Za, $Zb) = $polarization eq 's' # Q,w
    ? ($q/$k0, $q/$ka, $q/$kb) 
        : ($k0/($ceps0*$q), $ka/($cepsa*$q), $kb/($cepsb*$q));

the transfer matrices (Eqs. (15)),
my ($Ma, $Mb) = map { # n,m,Q,w
    my ($kd, $Z) = @$_;
    pdl([[cos($kd), i()*$Z*sin($kd)], [i()/$Z*sin($kd), cos($kd)]])
        ->mv(-1,0)->mv(-1,0); #m,n,Q,w
} ([$ka*$da, $Za], [$kb*$db, $Zb]);

the total transfer matrix (Eq. (16)),
my $M = $Mb x $Ma; # n,m,Q,w

and the eigenvalue (Eq. (19)).
my $cospd = ($M((0),(0)) + $M((1),(1)))/2; #Q,w
my $eipd = eipd($cospd); # Q,w

To get the eigenvalue I write a function eipd that takes the cosine and builds
the appropriate complex exponential. I use again thread_define so that I
write the routine for a scalar but then iterate implicitly with the threading
machine.

sub eipd {
    my $cospd = shift;
    my $eipd = null;
    _eipd($cospd, $eipd);
    return $eipd;
}

BEGIN {
    thread_define '_eipd(c();[o]r()));=> over {
        my ($c, $r) =@_; 
        my $s = sqrt(1-$c**2);
        my $eipd = $c + $s*i();
        my $mod2 = $eipd->abs2;
        $eipd = $c - $s*i() if $mod2 > 1;
        $eipd = $c - $s*i() if $mod2 == 1 && $s->re < 0; # ??
        $r.=$eipd;
    }
}
With the transfer matrix and its eigenvalue I can finally calculate the surface impedance and the reflection amplitude (give or take a sign, Eqs. (9))

# Notice that $M_{22}$->$M((1),(1))$ and $M_{21}$->$M((0),(1))$
# as the first index of a pdl designates columns, not rows
my $Z=($eipd-$M((1),(1)))/$M((0),(1)); # Q,w
my $r=($Z0-$Z)/($Z0+$Z); # Q,w

Finally, I plot the result. As I normalized distances with $d$ and took $c = 1$, my $\$variable $Q$ actually means $Qd$ and $\omega$ is actually $\omega d/c$.

4 Results

I run the program to get the magnitude of the reflection amplitude as a function of $Q$ and $\omega$.

4.1 Transverse electric

```
./reflection.pl --eps0 1 --epsa 2 --epsb 3 --im 1e-2 --da .5
--Qmax 10 --NQ 500 --wmax 10 --Nw 500 --pol te
--filename te
--title "Transverse electric e0=1 ea=2 eb=3 da=.5 db=.5 im=1e-2"
--cbrange "[0,5]"
```

4.2 Transverse magnetic

```
./reflection.pl --eps0 1 --epsa 2 --epsb 3 --im 1e-3 --da .5
--Qmax 10 --NQ 500 --wmax 10 --Nw 500 --pol tm
--filename tm
--title "Transverse magnetic e0=1 ea=2 eb=3 da=.5 db=.5 im=1e-3"
--cbrange "[0,5]"
```
Figure 2: Reflection amplitude for transverse electric polarization, for a system with $\epsilon_0 = 1$, $\epsilon_a = 2$, $\epsilon_b = 3$, $d_a = 0.5$ and $d_b = 0.5$. 
Figure 3: Reflection amplitude for transverse magnetic polarization, for a system with $\epsilon_0 = 1$, $\epsilon_a = 2$, $\epsilon_b = 3$, $d_a = 0.5$ and $d_b = 0.5$. 
4.3 Discussion

As can be seen from the plots, there is a nice sequence of Tamm modes. We reduced the cbrange scale to make the structure visible, but I expect the field amplification may be as large as the inverse of the fictitious imaginary part I added to the $\epsilon$'s. As the modes seem to be narrow, I do a closeup below.

```plaintext
./reflection.pl --eps0 1 --epsa 2 --epsb 3 --im 1e-3 --da .5 \
    --Qmin 5 --Qmax 6 --NQ 500 \
    --wmin 5 --wmax 6 --Nw 500 \
    --pol te --filename te_cu \
    --title "Transverse electric e0=1 ea=2 eb=3 da=.5 db=.5 im=1e-3" \
    --cbrange "[0,10]"
```

Figure 4: Reflection amplitude for transverse electric polarization, for a system with $\epsilon_0 = 1$, $\epsilon_a = 2$, $\epsilon_b = 3$, $d_a = 0.5$ and $d_b = 0.5$.

```plaintext
./reflection.pl --eps0 1 --epsa 2 --epsb 3 --im 1e-3 --da .5 \
    --Qmin 7 --Qmax 8 --NQ 500 \
```

14
Figure 5: Reflection amplitude for transverse magnetic polarization, for a system with $\epsilon_0 = 1$, $\epsilon_a = 2$, $\epsilon_b = 3$, $d_a = 0.5$ and $d_b = 0.5$.

5 Summary

Through a simple transfer matrix formalism we obtained the transfer matrix and from it the surface impedance, reflection amplitude and surface modes of a dispersionless dielectric superlattice. We found Tamm modes for both TE and TM polarizations, though the modes seem to couple more strongly in the TE than in the TM case, i.e., better tuning would be required to excite them in the TE case, though the amplification of the field is dominated by the artificial imaginary part added to the response and without dissipation it could reach infinity. The modes only exist outside of the light cone of the ambient, as could have been expected, and arise from the photonic bandgaps.
of the superlattice. We still have to explore how the dispersion relations depend on tunable parameters of the system such as the relative widths of the layers and their indices of refraction.

6 Appendix

6.1 Metapost code

I include below the metapost code used to produce Fig.

```
g_use_svg = 0; %0 for postscript, 1 for svg
prologues:=3;
u=1cm;
outputtemplate := "%j-%c.eps";
if g_use_svg > 0:
  outputtemplate := "%j-%c.svg";
  outputformat := "svg";
fi
beginfig(1)
for i=0 upto 4:
  fill unitsquare xscaled 5u yscaled .5u shifted (0,i*u)
    withcolor .5green+.5white;
  fill unitsquare xscaled 5u yscaled .5u shifted (0,.5u)
    shifted (0,i*u) withcolor .5blue+.5white;
  drawarrow((0,-.125u)--(.7u,.125u)) shifted(-.7u,0)
    shifted(-.2u,0) shifted (2.5u,0) shifted (0,.25u)
    shifted(0,i*u);
  drawarrow((0,.125u)--(.7u,-.125u))
    shifted(.2u,0) shifted (2.5u,0) shifted (0,.25u)
    shifted(0,i*u);
  drawarrow((0,-.20u)--(.7u,.20u)) shifted(-.7u,0)
    shifted(-.2u,0) shifted (2.5u,0) shifted (0,.75u)
    shifted(0,i*u);
  drawarrow((0,.20u)--(.7u,-.20u))
    shifted(.2u,0) shifted (2.5u,0) shifted (0,.75u)
    shifted(0,i*u);
endfor;
drawarrow((0,-.1u)--(.7u,.1u)) shifted(-.7u,0)
  shifted(-.2u,0) shifted (2.5u,0) shifted (0,-.2u);
```

16
I run the metapost code as shown below, and it produces a postscript file named `notas-1.eps`.

```plaintext
mpost notas
```